

Module 22 Stabiliteit van evenwichten

Onderwerp	Stabiliteit van evenwichten van gewone differentiaalvergelijkingen.
Voorkennis	Gewone differentiaalvergelijkingen, lineaire algebra.
Expressies	MatrixExponential, DEplot
Bibliotheken	LinearAlgebra, plots, DEtools
Zie ook	Module 14, 16, 21

22.1 Stabiliteit: definities

Algemeen. In de volgende definities is \mathbf{f} een autonoom vectorveld op \mathbb{R}^n , dat wil zeggen: we beschouwen het stelsel differentiaalvergelijkingen

$$\dot{\mathbf{x}} = \mathbf{f}(\mathbf{x}). \quad (22.1)$$

We nemen aan dat de oorsprong een *evenwicht* is, dat wil zeggen: $\mathbf{f}(\mathbf{0}) = \mathbf{0}$. De baan van het vectorveld door een punt \mathbf{x} geven we aan met $\phi_{\mathbf{x}}(t)$.

De oorsprong is een *stabiel evenwicht* als voor alle $\varepsilon > 0$ een $\delta > 0$ bestaat zo dat voor alle \mathbf{x} met $|\mathbf{x}| < \delta$ geldt dat $|\phi_{\mathbf{x}}(t)| < \varepsilon$ voor alle $t > 0$.

In feite staat hier dat de banen door elk punt dicht genoeg bij de oorsprong in de buurt blijven van de oorsprong. Als deze banen niet alleen in de nabijheid blijven van de oorsprong, maar hier zelfs naar toe convergeren, spreken we van asymptotische stabiliteit:

De oorsprong is *asymptotisch stabiel* als de oorsprong stabiel is en bovendien voor \mathbf{x} in een omgeving van de oorsprong geldt dat

$$\lim_{t \rightarrow \infty} \phi_{\mathbf{x}}(t) = \mathbf{0}.$$

Als de situatie optreedt dat er in een hoe kleine omgeving van de oorsprong dan ook, er steeds minimaal één punt in die omgeving is, zodat de baan door dit punt van de oorsprong wegloopt, spreken we van een instabiel evenwicht:

De oorsprong heet *instabiel* als er een $\varepsilon > 0$ bestaat zo dat bij iedere $\delta > 0$ een \mathbf{x} gevonden kan worden met $|\mathbf{x}| < \delta$ en $|\phi_{\mathbf{x}}(t^*)| > \varepsilon$ voor zekere $t^* > 0$.

Lineaire stelsels. We noemen het stelsel (22.1) *lineair* als **f** lineair is; (22.1) heeft dan de vorm

$$\dot{\mathbf{x}} = A \mathbf{x}, \tag{22.2}$$

met A een constante $n \times n$ -matrix. Het punt $\mathbf{0}$ is het enige evenwichtspunt van (22.2). De baan door \mathbf{x} wordt voor een lineair stelsel gegeven door

$$\phi_{\mathbf{x}}(t) = e^{At} \mathbf{x}. \tag{22.3}$$

De $n \times n$ -matrix e^{At} heet de *evolutie-operator* van (22.2). Er geldt dat het evenwicht asymptotisch stabiel is als alle eigenwaarden van A een strikt negatief reëel deel hebben.

22.2 De evolutie-operator e^{At}

MatrixExp.. De Maple-procedure `MatrixExponential` uit de `LinearAlgebra`-bibliotheek kan worden gebruikt om e^{At} te berekenen.

Voorbeeldopgave

Gegeven het lineaire stelsel $\mathbf{x}' = A \mathbf{x}$ met

$$A = \begin{bmatrix} -1 & 1 \\ 2 & 0 \end{bmatrix}.$$

Bepaal de aard van het evenwichtspunt en teken een aantal banen in het fasevlak.

Voorbeeldsessie

```
> restart; with(LinearAlgebra);
> A := Matrix( [[-1,1],[2,0]] );
```

$$A := \begin{bmatrix} -1 & 1 \\ 2 & 0 \end{bmatrix}$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} 1 \\ -2 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} & -1 \\ 1 & 1 \end{bmatrix}$$

Er is dus een zadelpunt.

```
> Phi := MatrixExponential(A,t);
```

$$\Phi := \begin{bmatrix} \frac{2}{3}e^{(-2t)} + \frac{1}{3}e^t & \frac{1}{3}e^t - \frac{1}{3}e^{(-2t)} \\ \frac{2}{3}e^t - \frac{2}{3}e^{(-2t)} & \frac{1}{3}e^{(-2t)} + \frac{2}{3}e^t \end{bmatrix}$$

We bekijken nu eerst hoe we een ‘baan’ door het punt \mathbf{v} als geparametriseerde kromme kunnen maken, die geschikt is om met een plot-commando te tekenen.

```
> v := <3,4>;
```

$$v := \begin{bmatrix} 3 \\ 4 \end{bmatrix}$$

```
> b := Phi.v;
```

$$b := \begin{bmatrix} \frac{2}{3}e^{-2t} + \frac{7}{3}e^t \\ \frac{14}{3}e^t - \frac{2}{3}e^{-2t} \end{bmatrix}$$

Er moet nog een range voor t achter

```
> bs := op( convert(b, list) );
```

$$bs := \frac{2}{3}e^{-2t} + \frac{7}{3}e^t, \frac{14}{3}e^t - \frac{2}{3}e^{-2t}$$

```
> baan1 := [bs,t=1..2];
```

$$baan1 := \left[\frac{2}{3}e^{-2t} + \frac{7}{3}e^t, \frac{14}{3}e^t - \frac{2}{3}e^{-2t}, t = 1..2 \right]$$

Dat vatten we samen in één procedure:

```
> baan := x -> [ op(convert(Phi.x,list)), t=-2..2 ];
```

$$baan := x \rightarrow [op(convert(\Phi . x, list)), t = -2..2]$$

```
> baan(v);
```

$$\left[\frac{2}{3}e^{-2t} + \frac{7}{3}e^t, \frac{14}{3}e^t - \frac{2}{3}e^{-2t}, t = -2..2 \right]$$


We maken een rij punten waardoor we banen willen tekenen:

```
> Q := <-1,1>, <1,-1>, <1/2,1>, <-1/2,-1>, <1,0>, <0,1>, <-1,0>, <0,-1>;
```

```
> plotlijst := seq( baan(x), x=Q );
```

```
> plot( [plotlijst], view=[-2..2,-2..2], color=black, thickness=3 );
```

(zie figuur 45.)



FIGUUR 45. Faseportret van een lineair stelsel: zadelpunt

Toelichting

Uiteraard hadden we het faseportret ook kunnen tekenen met DEplot. Met de evolutie-operator $\Phi(t) = e^{At}$ hebben we echter direct de beschikking over de exacte oplossingen. Zo is $\Phi(t)\mathbf{b}$ de oplossing $(x_1(t), x_2(t))$ met beginwaarde $(x_1(0), x_2(0)) = (b_1, b_2)$. Deze kunnen we tekenen als geparametriseerde kromme (zie §9.2). \diamond

22.3 Het principe van gelineariseerde stabiliteit

Asymptotische stabiliteit en instabiliteit van een vectorveld kan worden geanalyseerd door naar de linearisatie van dit vectorveld te kijken.

De linearisatie van \mathbf{f} in de oorsprong is het vectorveld $\mathbf{x} \rightarrow [D_0\mathbf{f}]\mathbf{x}$ met $[D_0\mathbf{f}]$ de Jacobi-matrix van \mathbf{f} in de oorsprong.

Het *principe van gelineariseerde stabiliteit* zegt, dat in een omgeving van een evenwicht waarvan geen van de eigenwaarden van de Jacobi-matrix van \mathbf{f} op de imaginaire as ligt, de faseplaatjes van de niet-lineaire vergelijking en van de gelineariseerde vergelijking ongeveer gelijk zijn (*topologisch equivalent*). In het bijzonder geldt dat een evenwichtspunt van de niet-lineaire vergelijking asymptotisch stabiel is als de reële delen van alle eigenwaarden van de linearisatie in het evenwichtspunt negatief zijn. Als een van de eigenwaarden een positief reëel deel heeft, dan is het evenwichtspunt instabiel.

Voorbeeldopgave

Gegeven is het vectorveld

$$\begin{aligned} \frac{dx_1}{dt} &= \sin(x_1 + x_2), \\ \frac{dx_2}{dt} &= x_2. \end{aligned}$$

Bepaal de evenwichten, de linearisatie in de evenwichten, en de stabiliteit van de evenwichten.

Voorbeeldsessie

```
> restart; with(LinearAlgebra): with(DEtools):
> f1,f2 := sin(x1+x2),x2;
```

```
f1, f2 := sin(x1 + x2), x2
> solve({f1,f2},{x1,x2});
      {x2 = 0, x1 = 0}
```

Dit antwoord is incorrect. Het goede antwoord is $x_2 = 0$ en $x_1 = 0 + k\pi$, $k \in \mathbb{Z}$.

```
> A := VectorCalculus:-Jacobian( [f1,f2], [x1,x2] );
```

$$A := \begin{bmatrix} \cos(x1 + x2) & \cos(x1 + x2) \\ 0 & 1 \end{bmatrix}$$

```
> A1 := eval( A, {x1=0,x2=0} );
```

$$A1 := \begin{bmatrix} 1 & 1 \\ 0 & 1 \end{bmatrix}$$

```
> Eigenvectors(A1);
```

$$\begin{bmatrix} 1 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 0 \end{bmatrix}$$

We zien dat er twee eigenwaarden zijn met een positief reëel deel. Dit evenwicht is dus instabil. Het is een instabiele ontaarde knoop. Het faseplaatje krijgen we als volgt (zie §21.6):

We maken eerst de linearisatie:

```
> rechterlid := A1.<x(t),y(t)>;
```

$$\text{rechterlid} := \begin{bmatrix} x(t) + y(t) \\ y(t) \end{bmatrix}$$

```
> stelsel := {diff(x(t),t)=rechterlid[1],
              diff(y(t),t)=rechterlid[2]};
```

$$\text{stelsel} := \left\{ \frac{d}{dt} x(t) = x(t) + y(t), \frac{d}{dt} y(t) = y(t) \right\}$$

```
> DEplot( stelsel, [x(t),y(t)], -5..2,
          {[0,0,1/2],[0,0,2],[0,0,-1/2],[0,0,-2],[0,1,0],[0,-1,0]},
          stepsize=0.1, scene=[x,y], x=-2..2, y=-2..2,
          title="Instabiele ontaarde knoop", arrows=smalltwo,
          color=red, linecolor=black, dirgrid=[12,12],
          axes=none );
```

Zie figuur 46, links)

Linearisatie rond het evenwichtspunt $(\pi, 0)$:

```
> A2 := eval( A, {x1=Pi,x2=0} );
```

$$A2 := \begin{bmatrix} -1 & -1 \\ 0 & 1 \end{bmatrix}$$

```
> Eigenvectors(A2);
```

$$\begin{bmatrix} 1 \\ -1 \end{bmatrix}, \begin{bmatrix} -\frac{1}{2} & 1 \\ 1 & 0 \end{bmatrix}$$

De linearisatie in dit evenwicht is een zadelpunt.


We doen nu het gelineariseerde stelsel 'uit het hoofd':

```
> stelsel := {diff(x(t),t)=-x(t)-y(t), diff(y(t),t)=y(t)};
```

$$\text{stelsel} := \left\{ \frac{d}{dt} y(t) = y(t), \frac{d}{dt} x(t) = -x(t) - y(t) \right\}$$

```
> DEplot( stelsel, [x(t),y(t)], -3..4,
          {[0,0,2],[0,0,-2],[0,-1,1],[0,1,-1],
           [0,1,-2],[0,-1,2],[0,1,0],[0,-1,0]},
          stepsize=0.1, scene=[x,y], x=-3..3, y=-3..3,
          title="Zadelpunt", dirgrid=[12,12], arrows=smalltwo,
          color=red, linecolor=black, axes=none );
```

(zie figuur 46, rechts)



FIGUUR 46. Linearisaties rond de evenwichtspunten van het stelsel $\mathbf{x}' = (\sin(x_1 + x_2), x_2)$

Toelichting

Het stelsel heeft dus instabiele ontaarde knopen in de punten $(2k\pi, 0)$ ($k \in \mathbb{Z}$) en zadelpunten in de punten $((2k+1)\pi, 0)$ ($k \in \mathbb{Z}$). Zie verder opgave 22.2. \diamond

22.4 Lyapunov-stabiliteit

Het principe van gelineariseerde stabiliteit kan niet worden gebruikt in het geval dat één van de eigenwaarden van de Jacobi-matrix van het vectorveld op de imaginaire as ligt. Hiervoor kunnen we soms gebruikmaken van een andere techniek. We zullen daarom nu toewerken naar enkele stellingen die stabiliteit van evenwichten karakteriseren met behulp van eigenschappen van een functie gedefinieerd op het vectorveld. Hiervoor hebben we nog enige voorbereidende definities nodig.

Een functie V heet positief definit (respectievelijk positief semi-definit) in een omgeving U van de oorsprong als $V(\mathbf{0}) = 0$ en $V(\mathbf{x}) > 0$ (respectievelijk $V(\mathbf{x}) \geq 0$) voor $\mathbf{x} \in U$ met $\mathbf{x} \neq \mathbf{0}$.

Een functie V heet negatief (semi-) definitief als $-V$ positief (semi-) definitief is.

Het is een bekende stelling dat de afgeleide van V in de richting van het vectorveld \mathbf{f} in het punt \mathbf{x} gelijk is aan het inproduct $\nabla V \cdot \mathbf{f}(\mathbf{x})$. Dit wordt meestal geschreven als $\frac{dV}{dt}$, ofwel \dot{V} .

In de straks te presenteren stellingen over de stabiliteit van evenwichten speelt het begrip Lyapunov-functie een hoofdrol: Een continu differentieerbare functie V heet een sterke (respectievelijk zwakke) *Lyapunov-functie* in een omgeving van de oorsprong als V positief definitief is en $\frac{dV}{dt}$ negatief definitief (respectievelijk negatief semi-definitief) is in die omgeving van de oorsprong.

Nu volgen de genoemde stellingen over de stabiliteit van evenwichten:

- (1) Als een sterke Lyapunov-functie bestaat in een omgeving van de oorsprong, dan is de oorsprong asymptotisch stabiel (Stelling van Lyapunov).
- (2) Als een zwakke Lyapunov-functie bestaat in een omgeving van de oorsprong, dan is de oorsprong stabiel (Stelling van Lyapunov).
- (3) Als een zwakke Lyapunov-functie bestaat in een omgeving van de oorsprong en de verzameling waar $\frac{dV}{dt} = 0$ geen andere baan bevat van het vectorveld dan de oorsprong, dan is de oorsprong asymptotisch stabiel.

Voorbeeldopgave

Laat zien dat de oorsprong van het vectorveld

$$\begin{aligned} \frac{dx}{dt} &= -y - x^3, \\ \frac{dy}{dt} &= x - x^2 y \end{aligned}$$

stabiel is.

Voorbeeldsessie

```
> restart;
> f1,f2 := -y-x^3, x-x^2*y;
      f1, f2 := -y - x^3, x - x^2 y
```

Als Lyapunov-functie proberen we $V(x, y) = x^2 + y^2$. We berekenen de afgeleide van V langs het vectorveld:

```
> V := x^2+y^2;
      V := x^2 + y^2
```

```
> V_dot := factor( diff(V,x)*f1 + diff(V,y)*f2 );
```

$$V_dot := -2x^2(x^2 + y^2)$$

We zien dat $\frac{dV}{dt}$ negatief semi-definiet is. De verzameling $x = 0$ bevat alleen de oorsprong zelf als baan. Dit kunnen we inzien als we het vectorveld bekijken op deze verzameling:

```
> subs( {x=0}, [f1,f2] );
```


$$[-y, 0]$$

Deze vector raakt alleen aan de verzameling $x = 0$ als ook $y = 0$. We concluderen dat de oorsprong asymptotisch stabiel is. In feite wordt het hele vlak door de oorsprong aangetrokken, omdat de condities voor een sterke Lyapunov-functie in het hele vlak gelden.

```
> with(DEtools):
```

```
> DEplot( {diff(x(t),t)=-y(t)-x(t)^3, diff(y(t),t)=x(t)-x(t)^2*y(t)},
[x,y], t=0..8, x=-3..3, y=-3..3,
{[0,3],[0,3,-3],[0,-3,-3],[0,-3,3]},
stepsize=0.05, arrows=SLIM, dirgrid=[12,12], linecolor=black );
```

(zie figuur 47.)



FIGUUR 47. Fasestroming van het stelsel $x' = -y - x^3$, $y' = x - x^2y$

Opgave 22.1

De differentiaalvergelijking

$$\frac{dx}{dt} = \mu x + y \quad \frac{dy}{dt} = -x + \mu y$$

hangt af van de parameter μ . Onderzoek hoe het faseplaatje verandert als μ verandert. Welke essentieel verschillende faseplaatjes kunt u onderscheiden, en met welke waarden van μ corresponderen die?

Geef ook de samenhang met de eigenwaarden van de matrix

$$\begin{bmatrix} \mu & 1 \\ -1 & \mu \end{bmatrix}.$$

Opgave 22.2

Maak een tekening van de fasestroming van de voorbeeldopgave van §22.3. Neem x_1 in het interval $[-4, 4]$ zodat er drie evenwichtspunten zichtbaar zijn.

Opgave 22.3

De differentiaalvergelijking

$$\begin{aligned} \frac{dx}{dt} &= \mu x - x(x^2 + y^2) \\ \frac{dy}{dt} &= \mu y - y(x^2 + y^2) \end{aligned}$$

hangt af van de parameter μ . Onderzoek hoe het faseplaatje verandert als μ verandert.

Opgave 22.4

Bepaal de evenwichten en gebruik het principe van gelineariseerde stabiliteit om, waar mogelijk, ze te karakteriseren.

- (a) $\frac{dx}{dt} = y^2 - 3x + 2$, $\frac{dy}{dt} = x^2 - y^2$
- (b) $\frac{dx}{dt} = y$, $\frac{dy}{dt} = -y + x^3$
- (c) $\frac{dx}{dt} = -y + x + xy$, $\frac{dy}{dt} = x - y - y^2$
- (d) $\frac{dx}{dt} = y$, $\frac{dy}{dt} = -(1 + x^2 + x^4)y - x$
- (e) $\frac{dx}{dt} = -3y + xy - 4$, $\frac{dy}{dt} = y^2 - x^2$

Opgave 22.5

Lineariseer het stelsel $\frac{dx}{dt} = -y$, $\frac{dy}{dt} = x - x^5$ in de oorsprong en zeg van welk type het evenwicht is. Laat zien dat de banen van de niet-lineaire vergelijking liggen op de familie van curven: $x^2 + y^2 - x^6/3 = C$, waar C een constante is. Schets deze curven om te laten zien dat de niet-lineaire vergelijking en de linearisatie hetzelfde faseplaatje hebben. Waarom kan dit niet uit het principe van gelineariseerde stabiliteit worden afgeleid?

Opgave 22.6

Laat zien dat $V(x, y) = x^2 + y^2$ een sterke Lyapunov-functie is in de oorsprong voor ieder van de volgende stelsels differentiaalvergelijkingen.

Wat kunt u zeggen over het aantrekkingsgebied? Probeer het numeriek of analytisch.

- (a) $\frac{dx}{dt} = -y - x^3, \quad \frac{dy}{dt} = x - y^3$
- (b) $\frac{dx}{dt} = -x^3 + y \sin(x), \quad \frac{dy}{dt} = -y - x^2 y - x \sin(x)$
- (c) $\frac{dx}{dt} = -x - 2y^2, \quad \frac{dy}{dt} = 2xy - y^3$
- (d) $\frac{dx}{dt} = -x \sin(x)^2, \quad \frac{dy}{dt} = -y - y^5$
- (e) $\frac{dx}{dt} = -(1 - y)x, \quad \frac{dy}{dt} = -(1 - x)y$

Opgave 22.7

Laat zien dat $V(x, y) = x^2 + y^2$ een zwakke Lyapunov-functie is in de oorsprong voor ieder van de volgende stelsels differentiaalvergelijkingen. Wat kunt u zeggen over het aantrekkingsgebied? Probeer het numeriek of analytisch.

- (a) $\frac{dx}{dt} = y, \quad \frac{dy}{dt} = -x - y^3(1 - x^2)^2$
- (b) $\frac{dx}{dt} = -x + y^2, \quad \frac{dy}{dt} = -xy - x^2$
- (c) $\frac{dx}{dt} = -x^3, \quad \frac{dy}{dt} = -x^2 y$
- (d) $\frac{dx}{dt} = -x + 2xy^2, \quad \frac{dy}{dt} = -x^2 y^3$

Opgave 22.8

Bewijs dat $V(x, y) = ax^2 + 2bxy + cy^2$ positief definit is dan en slechts dan als $a > 0$ en $b^2 < ac$. Gebruik dit om te laten zien dat de oorsprong van het stelsel

$$\frac{dx}{dt} = y, \quad \frac{dy}{dt} = -x - y + (x + 2y)(y^2 - 1)$$

asymptotisch stabiel is. Beschouw het gebied $|y| < 1$. Bepaal het attractiegebied. Schets het faseplaatje.

Opgave 22.9

Gegeven is het stelsel

$$\frac{dx}{dt} = y, \quad \frac{dy}{dt} = y - x^3.$$

Laat zien dat een positief definitie functie van de vorm $V(x, y) = ax^4 + bx^2 + cxy + dy^2$ kan worden gekozen zo dat ook $\frac{dV}{dt}$ positief definit is. Gebruik dit om aan te tonen dat de oorsprong instabiel is. Schets ook het faseplaatje in een omgeving van de oorsprong.